

Labour force participation

Age Cohorts	Younger
Surveys	Surveys 1 & 2
Derived Variables	WorkStdY
Definition	Labour force participation
Source Items (Index Numbers)	Survey 1: Questions 57 & 77 (EMPL-005 & DEMO-005) Survey 2: Question 75, parts a-g (EMPL-018 to -020 & EMPL-036 and TIME-024 to -026)
Statistical Form	Categorical variable
Index Number	EMPL-043
Prepared by	Sandra Bell and Anne Russell
Endorsed	13 March 2002

Source items - Survey 1

57 Which of the following BEST describes your MAIN current employment status? If you are studying AND working, circle the number corresponding to your MAIN activity.

Code	Response
1	In full time paid work
2	In part time or casual paid work
3	Work without pay (eg in a family business)
4	Home duties only – no paid work
5	Studying
6	Unemployed – looking for work
7	Unpaid voluntary work
8	Unable to work due to sickness or injury
9	Other (Please specify on line)

77 Are you currently attending an educational institution?

Code	Response
1	No
2	Yes, part-time student
3	Yes, full-time student

Source items - Survey 2

75 In the **LAST WEEK**, how much time in total did you spend doing the following things?

(Mark one on each line)

- a. Full-time paid work
- b. Permanent part-time paid work
- c. Casual paid work
- d. Home duties (own/family home)
- e. Work without pay (e.g. family business)
- f. Studying
- g. Unpaid voluntary work

Code	Response
0	I don't do this activity
1	1-15 hours
2	16-24 hours
3	25-34 hours
4	35-40 hours
5	41-48 hours
6	49 hours or more

Derived Variable

Work and study status at *Survey 1* was based on current employment status and attendance at an educational institution. Categories are:

Code	Category
1	Studies, no work
2	Works, no study
3	Works and studies
4	At home
5	No work
6	Other

At *Survey 2* work and study status was defined for all women in the Younger cohort who completed the full or abbreviated version of the survey. The definition is based on:

- the total number of hours worked;
- whether women study; and
- whether women are occupied with home duties and or unpaid voluntary work.

The number of hours worked is calculated as the sum of the coded responses to items a, b, c and e (full time paid work, permanent part-time paid work, casual paid work, work without pay). Values are given the same interpretation as the coded responses to individual items. Total hours work of 35 or more are characterized as full-time work; 34 or less are characterized as part-time work.

Table 1 shows the number and percent of women in each category at Surveys 1 and 2, where the categories are defined as:

Code	Category
1	Studies, no work <i>Total hours of work is zero and hours of study is 1 or more</i>
2	Works Part Time, no study <i>Total hours of work is 1-34 and hours of study is zero</i>
3	Works Full Time, no study <i>Total hours of work is 35 or more and hours of study is zero</i>
4	Works Part Time and studies <i>Total hours of work is 1-34 and hours of study is 1 or more</i>
5	Works Full Time and studies <i>Total hours of work is 35 or more and hours of study is 1 or more</i>
6	No Work, No Study <i>Total hours of work is zero and hours of study is zero</i>

Table 1 **Number and percent in each work-study category**

Work-study status	Survey 1		Survey 2	
	Number	Percent	Number	Percent
Survey 1	14 777		9 685	
Studies, no work	4 618	31.9	372	3.9
Works, no study	5 773	39.9	5 819	60.3
<i>Works Part Time, no study</i>			1 490	15.4
<i>Works Full Time, no study</i>			4 329	44.9
Works and studies	1 929	13.3	2 342	24.3
<i>Works Part Time and studies</i>			1 020	10.6
<i>Works Full Time and studies</i>			1 322	13.7
No Work, no study ¹	1 147	14.9	1 113	11.5
At home	910	6.3		
No work	1 007	7.0		
Other	230	1.6		
Total Classified	14 467	100.0	9 646	100.0
<i>Missing</i>	312	2.1	39	0.4

¹ Comprises 1 061 women reporting time spent in home duties and/or voluntary work and 52 women reporting no activity for work, study, home duties or voluntary work

The SAS code defining work and study status at surveys 1 and 2 is:

```
Survey 1
  if y1q57=5 and y1q77 in (2,3) then y1workStudy=1;
else if y1q57 in (1,2,3) then do ;
  if y1q77 in (2,3) then y1workStudy=2;
  else if y1q77 in (1,.) then y1workStudy=3;
end ;
else if y1q57=4 then do ;
  if y1q77 in (2,3) then y1workStudy=1;
  else if y1q77 in (1,.) then y1workStudy=4;
end ;
else if y1q57 in (6,7,8) then do ;
  if y1q77 in (2,3) then y1workStudy=1;
  else if y1q77 in (1,.) then y1workStudy=5;
end ;
else if y1q57=9 then y1workStudy=6;

Survey 2
array emp {4} y2q75a y2q75b y2q75c y2q75e;
array q75 {7} y2q75a y2q75b y2q75c y2q75d y2q75e y2q75f y2q75g ;
/*****
Calculate 'hours employed'
*****/
emphours = sum (of emp{*}) ;
if emphours > 6 then emphours = 6;
if emphours = 0 and nmiss (of emp {*}) >=1 then emphours = . ;

if emphours >0 then do;
  if y2q75f in (.,0) then do;
    if emphours <=3 then y2workStudy = 2;
    else if emphours >3 then y2workStudy = 3;
  end;
  else if y2q75f ge 1 then do;
    if emphours <=3 then y2workStudy = 4;
    else if emphours > 3 then y2workStudy = 5;
  end;
end ;
else if emphours in (.,0) then do ;
  if y2q75f ge 1 then y2workStudy = 1;
  else if y2q75f in (.,0) and
    (y2q75d >=1 or y2q75g >=1) then y2workStudy=6;
  else if sum (of q75 {*}) = 0 and nmiss (of q75{*)=0
  then y2workStudy = 6;
end;
```

Age Cohorts	Mid-age
Surveys	Survey 1, 2 & 3
Derived Variables	LabF
Definition	Labour force participation
Source Items (Index Numbers)	Survey 1: Question 66 (EMPL-001) Survey 2: Question 54 (Main & secondary) (EMPL-006 & EMPL-007) Survey 3: Question 71 a-h (EMPL-018 to -020; TIME-024 to -027 & TIME-029)
Statistical form	Categorical variable
Index Number	EMPL-042
Derived Variable	Paid
Definition	Payment for work
Source Items (Index Numbers)	Survey 1: Question 66 & 67 (EMPL-001 & TIME-001)) Survey 2: Question 54 (Main & secondary) & Question 55 (EMPL-006 to EMPL-008) Survey 3: Question 71 a-h (EMPL-018 to -020; TIME-024 to -027 & TIME-029)
Statistical form	Categorical variable
Index Number	EMPL-044
Derived Variable	HrsWork
Definition	Hours worked
Source Items (Index Numbers)	Items and Indexes are the same as M1Paid above
Index Number	EMPL-041
Statistical Form	Categorical variable
Prepared by	Anne Russell and Jessica Ford
Endorsed	19 March 2003

Source items - Survey 1

66 Which of the following best describes your main current employment status?

Code	Response
1	In full time paid work
2	In part time or casual paid work
3	Work without pay (eg in a family business)
4	Home duties only - no paid work
5	Studying - no paid work
6	Unemployed - looking for work
7	Unpaid voluntary work
8	Retired
9	Unable to work due to sickness or injury
10	Other

67 How many hours do you normally spend in all your PAID jobs each week?

Code	Response
1	1 - 15 hours
2	16 - 24 hours
3	25 - 34 hours
4	35 - 40 hours
5	41 - 48 hours
6	49 hours or more
7	Not in paid work

Source items - Survey 2

54 Which of the following describes your MAIN and SECONDARY occupation status? (Eg if you work part-time and do home duties, the one which takes most time would be MAIN and the other would be SECONDARY). (Mark one circle in each column)

Code		Response
Main	Secondary	
1	1	In full time paid work
2	2	In part time or casual paid work
3	3	Work without pay (eg in a family business)
4	4	Home duties
5	5	Studying
6	6	Unemployed - looking for work
7	7	Unpaid voluntary work
8	8	Retired
9	9	Unable to work due to sickness or injury
	0	No secondary occupation

55 How many hours do you normally spend in your paid work each week?

Code	Response
1	None
2	1 - 15 hours
3	16 - 24 hours
4	25 - 34 hours
5	35 - 40 hours
6	41 - 48 hours
7	49 hours or more

Source items - Survey 3

71 In the LAST WEEK, how much time in total did you spend doing the following things? (Mark one on each line)

- a) Full-time paid work
- b) Permanent part-time paid work
- c) Casual paid work
- d) Home duties (own/family home)
- e) Work without pay (eg family business)
- f) Looking for work
- g) Unpaid voluntary work
- h) Studying

Code	Response
0	I don't do this activity
1	1-15 hours
2	16-24 hours
3	25-34 hours
4	35-40 hours
5	41-48 hours
6	49 hours or more

Labour Force Definitions

The ABS framework¹ for labour force participation has been used. According to his framework members of the population are classified into one of three mutually exclusive categories:

- In the labour force – Employed (works for pay, profit or commission; or works in a family business or on a farm)
- In the labour force – Unemployed (Both looking for and available for work)
- Not in the labour force (includes home duties, retired, voluntarily inactive, unable to work, persons in institution, unpaid voluntary work)

Employed persons can be further categorised with respect to payment for work:

- Paid workers are in full-time, part-time or casual paid employment;
- Unpaid workers are employed without pay in a family business or on a farm (known as contributing family workers in the ABS framework).

Employed persons can also be further categorised with respect to the number of hours worked:

- 1-34 hours (consistent with part-time work)
- 35 hours or more (consistent with full-time work)

Derived Variables

Based on the ABS framework, 3 variables have been created from each survey of the Mid-age cohort to describe employment:

- Labour force participation: M1LabF, M2LabF and M3LabF;
- Payment for work: M1Paid, M2Paid and M3Paid and
- Hours worked: M1HrsWork, M2HrsWork and M3HrsWork.

The survey items used to define labour force participation and payment for work were available for women completing Survey 1, the full and short versions of Survey 2 and the full version only of Survey 3.

Classification for labour force participation at Survey 1 was based on a single item concerning main current employment and at Survey 2 was based on a similar item including both main and secondary occupation. At Survey 3, the 7 items within question 71 recorded the hours spent in various occupational activities. Women were defined as being involved in any activity if they reported 1 or more hours spent on that activity.

The number and percent of women in each category of labour force participation and payment for work is shown in Table 1a.

Table 1a Number and percent of women in each category of labour force participation and payment for work

	Mid-age 1		Mid-age 2		Mid-age 3	
	Number	%	Number	%	Number	%
Surveys for which source items are available	Full		Full & short		Full	
Completed surveys	14 099		12 338		11 196	
<i>Not classified</i>	181		210		73	
Classified for:	M1LabF		M2Labf		M3Labf	
In the labour force-Employed	10 251	73.6	9 531	78.6	8 580	77.1
In the labour force-Unemployed	261	1.9	183	1.5	147	1.3
Not in the labour force	3 406	24.5	2 414	19.9	2 396	21.5
Total classified	13 918	100.0	12 128	100.0	11 123	100.0
		0				
Classified for:	M1Paid		M2Paid		M3Paid	
Employed - Paid	9 173	65.9	8 592	70.8	7 309	65.7
Employed - Not paid	1 078	7.7	939	7.7	1 271	11.4
Not in the labour force or unemployed	3 667	26.4	2 597	21.4	2 543	22.9
Total classified	13 918	100	12 128	100	11 123	100

The survey items used to define the number of hours worked were available for women completing Survey 1 and the full version of surveys 2 and 3. Classification for hours worked was based on a single item at surveys 1 and 2. At Survey 3, hours worked was calculated as the total time reported for 4 activities

- full-time paid work (M3Q71a)
- permanent part-time paid work (M3Q71b)
- casual paid work (M3Q71c)
- work without pay, eg. family business (M3Q71e)

Responses for each activity were assigned an estimated number of hours (mostly the median for the response class) and a total calculated. The total was categorized according to the response categories for individual items. The estimate of 8 hours for the category '1-15 hours' ensures that where there are two responses in this category, the total is 16 and categorised as '16-24 hours'.

Response	Estimated hours
Don't do this activity	0
1-15 hours	8
16-24 hours	20
25-34 hours	29.5
35-40 hours	37.5
41-48 hours	44.5
49+ hours	49

The number and percent of women in each category of hours worked are shown in Table 2. Differences between surveys in the base data from which hours worked were derived have resulted in higher rates of missing data at surveys 1 & 2 than at Survey 3. These differences arise mainly due to data deficiencies in relation to the hours worked without payment in the first 2 surveys.

Rates of missing data differed among employed women who are paid and those who are not (Table 3). There were no unpaid, employed women who could be classified for hours worked at survey1, while 69% were classified at Survey 2 and 100% were classified at Survey 3.

Table 2 Number and percent of women in each category for number of hour worked (Full survey data for surveys 1, 2 & 3)

	Mid-age 1		Mid-age 2		Mid-age 3	
	Number	%	Number	%	Number	%
Completed surveys	14 099		11 648		11 196	
<i>Not classified</i>	1 306	9.3	1 005	7.5	73	0.7
1-15 hours	1 337	10.5	1 140	10.7	1 364	12.3
16-24 hours	1 497	11.7	1 260	11.8	1 296	11.6
25-34 hours	1 587	12.4	1 492	14.0	1 312	11.8
35-40 hours	2 733	21.4	2 336	22.0	2 192	19.7
41-48 hours	1 250	9.8	1 175	11.0	1 179	10.6
49 hours or more	722	5.6	776	7.3	1 237	11.1
Not in the labour force or unemployed	3 667	28.9	2 464	23.2	2 543	22.9
Total classified	12 793	100.0	10 643	100.0	11 123	100.0

Table 3 Number and percent of women in paid and unpaid employment who were classified for number of hour worked (Full survey data for surveys 1, 2 & 3)

Hours of work	Mid-age 1		Mid-age 2		Mid-age 3	
	Not paid	Paid	Not paid	Paid	Not paid	Paid
<i>Not classified</i>						
Number	1 078	47	617	180	0	0
Percent	100.0	0.5	69.2	2.2	0.0	0.0
<i>Classified</i>						
Number	0	9 126	275	7 904	1 271	7 309
Percent	0.0	99.5	30.8	97.8	100.0	100.0
Total	1 078	9 173	892	8 084	1 271	7 309

There are other areas in which choices in categorisation may have resulted in differences between surveys. At surveys 2 and 3, when women could report more than 1 activity, women were first classified as being in paid employment if they reported any paid activity; only women reporting unpaid activity alone are classified as unpaid. So the 'paid' category includes women in paid employment only and both paid and unpaid employment. At Survey 2, 8% (448/7 904) of women


who were classified as being in paid employment also reported doing work without pay. At Survey 3, 21% (1 532/7 309) of women who were classified as being in paid employment also reported work without pay. This may reflect real change in employment between surveys or methodological differences. It may be preferable to consider women with a mixed work pattern (i.e. both paid and unpaid work) as a separate group in any statistical analysis.

The pattern of hours worked was significantly different for women in paid and unpaid work at surveys 2 and 3 ($p < 0.0001$ both surveys, Table 4), although the pattern for each group was similar at both surveys (Figure 1).

Table 4 Percent of women in paid and unpaid employment working various hours

Hours worked	Mid-age 2		Mid-age 3	
	Not paid	Paid	Not paid	Paid
1 to15 hours	34.6	13.2	44.2	11.0
16 to 24 hours	13.5	15.5	13.3	15.4
25 to 34 hours	18.9	18.2	10.2	16.2
35 to 40 hours	9.5	29.2	7.7	28.7
41 to 48 hours	4.4	14.7	6.9	14.9
49 hours or more	19.3	9.2	17.4	13.9

Figure 1 Percent of women in paid and unpaid employment working various hours



Generally, the analysis of payment for work and of hours worked will be restricted to employed women and hours worked will be summarised as to 1 to 34 hours or 35 hours or more (Table 5).

Table 5 Summary employment variables for Mid-age women (Number and percent)

	Survey 1		Survey 2		Survey 3	
	Number	%	Number	%	Number	%
'Employed' women	10 251		9 531		8 580	
Payment for work						
Employed - Paid	9 173	89.5	8 592	90.1	7 309	85.2
Employed - Not paid	1 078	10.5	939	9.9	1 271	14.8
Total	10 251	100.0	9 531	100.0	8 580	100.0
Hours worked						
1-15 hours	1 337	14.7	1 140	13.9	1 364	15.9
16-24 hours	1 497	16.4	1 260	15.4	1 296	15.1
25-34 hours	1 587	17.4	1 492	18.2	1 312	15.3
35-40 hours	2 733	29.9	2 336	28.6	2 192	25.5
41-48 hours	1 250	13.7	1 175	14.3	1 179	13.7
49 hours or more	722	7.9	776	9.5	1 237	14.4
Part-time	4 421	48.4	3 892	47.6	3 972	46.3
- 1 to 34 hours						
Full-time	4 705	51.6	4 287	52.4	4 608	53.7
-35 hours or more						
Total	9 126	100.0	8 179	100.0	8 580	100.0

The SAS code defining work related variables for mid-age women at Surveys 1, 2 and 3 is:

```

Survey 1
data mid1 ;
set phase1.wha1mid;
/*****
labour force participation
0=not in labour force (hd,study,unpaid vol,unable,retired)
1=in labour force employed (ft,pt,cas,wwop)
2=in labour force unemployed
*****/;
if m1q66 in (1,2,3) then m1labf = 1;
else if m1q66 = 6 then m1labf=2;
else if m1q66 in (4,5,7,8,9,10) then m1labf = 0;

/*****
Payment for work
0=Employed, not paid
1=Employed, paid
2='not in labour force/unemployed
*****/;
if m1labf = 1 then do;
  if m1q66 in (1,2) then m1paid = 1;
  else if m1q66 = 3 then m1paid = 0;
end;
else if m1labf in (2,0) then m1paid = 2;

/*****
number of hours worked
0=none          4=35-40
1=1-15         5=41-48
2=16-24        6=49+
3=25-34        7=not labf/unemp
*****/;
if m1q67=7 then m1q67= 0;

if m1labf=1 then do;
  if m1q67 = 0 then m1hrswork = .;
  else m1hrswork = m1q67;
end;
else if m1labf in (0,2) then m1hrswork=7;

/*****
collapse hours worked to part-time/full-time
1=1-34 hours (part-time)
2='35+ hours (full-time)
0='not in labour force/unemployed
*****/;
if m1hrswork in (1,2,3) then m1hrs = 1;
else if m1hrswork in (4,5,6) then m1hrs = 2;
else if m1hrswork = 7 then m1hrs = 0;
run;

Survey 2
data mid2 ;
set phase2.wha2mid;
/*****
labour force participation
0=not in labour force (hd,study,unpaid vol,unable,retired)
1=in labour force employed (ft,pt,cas,wwop)
2=in labour force unemployed
*****/;
  if m2survey = . then m2labf = . ;
else if m2q54mn=. and m2q54sc=. and m2q55n=. then m2labf=. ;
else if m2q54mn in (1,2,3) and m2q54sc in (.,0,1,2,3,4,5,6,7,8,9) then
  m2labf=1;
else if m2q54sc in (1,2,3) and m2q54mn in (.,1,2,3,4,5,6,7,8,9,0) then
  m2labf=1;
else if m2q54mn in (6) and m2q54sc in (.,0,1,2,3,4,5,6,7,8,9) then
  m2labf=2;
else if m2q54sc in (6) and m2q54mn in (.,1,2,3,4,5,6,7,8,9,0) then
  m2labf=2;
else if m2q54mn in (4,5,7,8,9) and m2q54sc in (.,0,4,5,7,8,9) then
  m2labf =0;
else if m2q54sc in (4,5,7,8,9) and m2q54mn in (.,0,4,5,7,8,9) then

```

```

m2labf =0;
/*****
Payment for work
0=Employed, not paid
1=Employed, paid
2=not in labour force/unemployed;
*****/;
  if m2labf = 1 then do;
    if m2q54mn in (1,2) or m2q54sc in (1,2) then m2paid = 1;
    else if m2q54mn = 3 or m2q54sc = 3 then m2paid = 0;
    end;
else if m2labf in (2,0) then m2paid = 2;

/*****
  recode survey 1 item 55 to have same codes as
  Survey 1 item 67 *****/;
if m2q55=1 then m2q55n = 0;
else m2q55n = m2q55 -1;
/*****
number of hours worked
0=none          4=35-40
1=1-15         5=41-48
2=16-24        6=49+
3=25-34        7=not labf/unemp
*****/;
if m2survey in (2,.) then m2hrswork = .;
else if m2labf=1 then do;
  if m2q55n = 0 then m2hrswork = .;
  else m2hrswork=m2q55n ;
end;
else if m2labf in (0,2) then m2hrswork=7;

/*****
collapse hours worked to part-time/full-time
1=1-34 hours (part-time)
2='35+ hours (full-time)
0='not in labour force/ unemployed
*****/;
if m2hrswork in (1,2,3) then m2hrs = 1;
else if m2hrswork in (4,5,6) then m2hrs = 2;
else if m2hrswork = 7 then m2hrs = 0;
run;

Survey 3
data mid3;
set phase3.wha3mid;
/*****
labour force participation
0=not in labour force (hd,study,unpaid vol,unable,retired)
1=in labour force employed (ft,pt,cas,wwop)
2=in labour force unemployed
*****/;
if m3q71a in (1,2,3,4,5,6) or m3q71b in (1,2,3,4,5,6) or
  m3q71c in (1,2,3,4,5,6) or m3q71e in (1,2,3,4,5,6) then m3labf=1;
else if m3q71f in (1,2,3,4,5,6) then m3labf = 2;
else if m3q71d in (1,2,3,4,5,6) or m3q71g in (1,2,3,4,5,6) or
  m3q71j in (1,2,3,4,5,6) then m3labf=0;
else if m3q71a=0 and m3q71b =0 and m3q71c =0 and m3q71e =0 and
  m3q71f=0 then m3labf = 0;
/*****
Payment for work
0=Employed, not paid
1=Employed, paid
2='not in labour force/unemployed' ;
*****/;
array paid {3} m3q71a m3q71b m3q71c;
array paidnew {3} m3q71an m3q71bn m3q71cn;

do i= 1 to 3;
  if paid{i} in (1,2,3,4,5,6) then paidnew{i} = 1;
  else if paid {i} = 0 then paidnew{i} = 0;
end;

if m3labf=1 then do;
  if sum (of paidnew{*)} >0 then m3paid = 1;
  else if sum (of paidnew{*)}=0 or m3q71e in (1,2,3,4,5,6)

```

```

then m3paid = 0;
end;
else if m3labf in (0,2) then m3paid=2;

/*****
- hrs paid work includes FT, PT, casual and work without pay */
- if employed then have hours worked
*****/;
array work {4} m3q71a m3q71b m3q71c m3q71e;
array workavg {4} avga avgb avgc avge;
do i = 1 to 4 ;
    if work{i} = 0 then workavg{i} = 0 ;
else if work{i} = 1 then workavg{i} = 8 ;
else if work{i} = 2 then workavg{i} = 20 ;
else if work{i} = 3 then workavg{i} = 29.5 ;
else if work{i} = 4 then workavg{i} = 37.5 ;
else if work{i} = 5 then workavg{i} = 44.5 ;
else if work{i} = 6 then workavg{i} = 49 ;
else if work{i} in (9,.) then workavg{i} = . ;
end ;

totavg = sum (of workavg{*});
numiss = nmiss (of workavg{*}) ;

/*****
number of hours worked
0=none          4=35-40
1=1-15         5=41-48
2=16-24       6=49+
3=25-34       7=not labf/unemp
*****/;
if m3labf = 1 then do;
    if numiss = 4 then m3hrswork = . ;
else if numiss=0 and totavg = 0 then m3hrswork = . ;
else if 1<=totavg<=15 then m3hrswork = 1;
else if 16<=totavg<=24 then m3hrswork = 2;
else if 25<=totavg<=34 then m3hrswork = 3;
else if 35<=totavg<=40 then m3hrswork = 4 ;
else if 41<=totavg<=48 then m3hrswork = 5 ;
else if totavg>=49 then m3hrswork = 6 ;
else m3hrswork = . ;
end;
else if m3labf in (0,2) then m3hrswork=7;

/*****
collapse hours worked to part-time/full-time
1=1-34 hours (part-time)
2='35+ hours (full-time)
0='not in labour force/ unemployed
*****/;
if m3hrswork in (1,2,3) then m3hrs = 1;
else if m3hrswork in (4,5,6) then m3hrs = 2;
else if m3hrswork = 7 then m3hrs = 0;
run;

```

Reference

1. Australian Bureau of Statistics. *Labour statistics: Concepts, sources and methods*. 2001 Australian Bureau of Statistics Catalogue No. 6102.0. Available from: [http://www.ausstats.abs.gov.au/ausstats/free.nsf/Lookup/819D9201B77786E7CA256AA0000667C3/\\$File/61020_2001.pdf](http://www.ausstats.abs.gov.au/ausstats/free.nsf/Lookup/819D9201B77786E7CA256AA0000667C3/$File/61020_2001.pdf) [cited 2003 January 23]